NAG Library Routine Document

G05HMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05HMF generates a given number of terms of a GJR GARCH(p, q) process (see Glosten *et al.* (1993)).

2 Specification

```
 SUBROUTINE G05HMF(DIST, NUM, IP, IQ, THETA, GAMMA, DF, HT, ET, FCALL,
1                  RVEC, IGEN, ISEED, RWSAV, IFAIL)
 INTEGER           NUM, IP, IQ, IGEN, ISEED(4), IFAIL
 double precision  THETA(IQ+IP+1), GAMMA, DF, HT(NUM), ET(NUM), RVEC(40),
1                  RWSAV(9)
 LOGICAL           FCALL
 CHARACTER*1       DIST
```

3 Description

A GJR GARCH(p, q) process is represented by:

$$h_t = \alpha_0 + \sum_{i=1}^{q}(\alpha_i + \gamma S_{t-i})\epsilon_{t-i}^2 + \sum_{i=1}^{p}\beta_i h_{t-i}, \qquad t = 1, 2, \ldots, T.$$

where $S_t = 1$, if $\epsilon_t < 0$, $S_t = 0$, if $\epsilon_t \geq 0$, and $\epsilon_t \mid \psi_{t-1} = N(0, h_t)$ or $\epsilon_t \mid \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t-distribution with df degrees of freedom and variance h_t, T is the number of observations in the sequence, ϵ_t is the *observed* value of the GARCH(p, q) process at time t, h_t is the conditional variance at time t, and ψ_t the set of all information up to time t. Symmetric GARCH sequences are generated when γ is zero, otherwise asymmetric GARCH sequences are generated with γ specifying the amount by which negative shocks are to be enhanced.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05HMF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER*1 *Input*

On entry: the type of distribution to use for ϵ_t.

DIST = 'N'
A Normal distribution is used.

DIST = 'T'
A Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: NUM – INTEGER *Input*

On entry: T, the number of terms in the sequence.

Constraint: NUM > 0.

3: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, 2, \ldots, p$.

Constraints:

IQ + IP + 1 $\leq$ 20;
IP $\geq$ 0.

4: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, 2, \ldots, q$.

Constraints:

IQ + IP + 1 $\leq$ 20;
IQ $\geq$ 1.

5: THETA(IQ + IP + 1) – ***double precision*** array *Input*

On entry: the first element must contain the coefficient α_o, the next IQ elements must contain the coefficients α_i, for $i = 1, 2, \ldots, q$. The remaining IP elements must contain the coefficients β_j, for $j = 1, 2, \ldots, p$.

6: GAMMA – ***double precision*** *Input*

On entry: the asymmetry parameter γ for the GARCH(p, q) sequence.

7: DF – ***double precision*** *Input*

On entry: the number of degrees of freedom for the Student's t-distribution.

If DIST = 'N', DF is not referenced.

Constraint: if DIST = 'T', DF > 2.

8: HT(NUM) – ***double precision*** array *Output*

On exit: the conditional variances h_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

9: ET(NUM) – ***double precision*** array *Output*

On exit: the observations ϵ_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

10: FCALL – LOGICAL *Input*

On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in RVEC.

11: RVEC(40) – ***double precision*** array *Input/Output*

On entry: the array contains information required to continue a sequence if FCALL = .FALSE..

On exit: contains information that can be used in a subsequent call of G05HMF, with FCALL = .FALSE..

12: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

13: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

14: RWSAV(9) – ***double precision*** array *Communication Array*

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, IP $<$ 0,
or IQ $<$ 1,
or DF $\leq$ 2,
or NUM $\leq$ 0,
or DIST $\neq$ 'N' and DIST $\neq$ 'T',
or IQ + IP + 1 $>$ 20.

IFAIL = 2

On entry, $\alpha_i + \gamma < 0$ for some α_i.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example uses G05HMF to generate 1500 data points, with known process parameters θ for the following two time series:

(i) a GARCH$(1,1)$ sequence with Normally distributed residuals;

(ii) a GARCH$(1,1)$ sequence with Student's t-distributed residuals.

Here G05HMF is initially called, with the output discarded, to eliminate 'start-up effects' in these sequences. The process parameter estimates, $\hat{\theta}$, are then obtained using G13FEF, and compared with their true values, θ. Finally a four step ahead volatility estimate is computed using G13FFF.

9.1 Program Text

```
*     G05HMF Example Program Text
*     Mark 22 Release. NAG Copyright 2008.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPARMX, LDCOVR, NUM
      DOUBLE PRECISION ZERO
      PARAMETER        (NPARMX=10,LDCOVR=NPARMX,NUM=2000,ZERO=0.0D0)
      INTEGER          NUM1, NREGMX
      PARAMETER        (NUM1=3000,NREGMX=10)
*     .. Local Scalars ..
      DOUBLE PRECISION DF, FAC1, GAMMA, HP, LGF, MEAN, TOL, XTERM
      INTEGER          I, IFLAG, IGEN, IP, IQ, K, LDX, LWORK, MAXIT, MN,
     +                 NPAR, NPAR2, NREG, NT
      LOGICAL          FCALL
      CHARACTER        DIST
*     .. Local Arrays ..
      DOUBLE PRECISION BX(10), COVR(LDCOVR,NPARMX), CVAR(100),
     +                 ETM(NUM1), HT(NUM1+10), HTM(NUM1), PARAM(NPARMX),
     +                 RVEC(40), RWSAV(9), SC(NPARMX), SE(NPARMX),
     +                 THETA(NPARMX), WORK(NUM1*3+NPARMX+NREGMX*NUM1+20*
     +                 20+1), X(NUM1,10), YT(NUM1+10)
      INTEGER          ISEED(4)
      LOGICAL          COPTS(2)
*     .. External Subroutines ..
      EXTERNAL         G05HMF, G05KBF, G13FEF, G13FFF
*     .. Intrinsic Functions ..
      INTRINSIC        DBLE, SIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05HMF Example Program Results'
      WRITE (NOUT,*)
*
      ISEED(1) = 111
      IGEN = 0
*
      LWORK = NUM1*3 + NPARMX + NREGMX*NUM1 + 1
      NREG = 0
      LDX = NUM1
      DF = 5.1D0
      GAMMA = 0.1D0
      BX(1) = 1.5D0
      BX(2) = 2.5D0
      BX(3) = 3.0D0
      MEAN = 4.0D0
*
      DO 20 I = 1, NUM
         FAC1 = DBLE(I)*0.01D0
         X(I,2) = 0.01D0 + 0.7D0*SIN(FAC1)
         X(I,1) = 0.5D0 + FAC1*0.1D0
         X(I,3) = 1.0D0
   20 CONTINUE
*
      MN = 1
      NREG = 2
```

```
      GAMMA = 0.1D0
      IP = 1
      IQ = 1
      NPAR = IQ + IP + 1
      PARAM(1) = 0.4D0
      PARAM(2) = 0.1D0
      PARAM(3) = 0.7D0
      FCALL = .TRUE.
*
      DIST = 'N'
      CALL G05KBF(IGEN,ISEED)
      IFLAG = 1
      CALL G05HMF(DIST,200,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
*
      IF (IFLAG.EQ.0) THEN
         FCALL = .FALSE.
         IFLAG = 1
         CALL G05HMF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,
     +               IGEN,ISEED,RWSAV,IFLAG)
      END IF
      IF (IFLAG.NE.0) THEN
         WRITE (NOUT,99997) IFLAG
         GO TO 300
      END IF
*
      DO 60 I = 1, NUM
         XTERM = ZERO
         DO 40 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
   40    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
   60 CONTINUE
*
      IFLAG = -1
*
      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.
*
      MAXIT = 100
      TOL = 1.0D-5
*
      DO 80 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5D0
   80 CONTINUE
      THETA(NPAR+1) = GAMMA*0.5D0
      IF (MN.EQ.1) THEN
         THETA(NPAR+MN+1) = MEAN*0.5D0
      END IF
      DO 100 I = 1, NREG
         THETA(NPAR+MN+1+I) = BX(I)*0.5D0
  100 CONTINUE
*
      NPAR2 = 2 + IP + IQ + MN + NREG
      CALL G13FEF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,NPAR2,THETA,SE,SC,
     +            COVR,LDCOVR,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WORK,LWORK,
     +            IFLAG)
*
      IF (IFLAG.LT.0) GO TO 300
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Normal distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '          Parameter       Standard        Correct'
      WRITE (NOUT,*) '          estimates       errors          values'
*
      DO 120 I = 1, NPAR
         WRITE (NOUT,99999) THETA(I), SE(I), PARAM(I)
```

```
  120 CONTINUE
*
      WRITE (NOUT,99999) THETA(NPAR+1), SE(NPAR+1), GAMMA
*
      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+2), SE(NPAR+2), MEAN
      END IF
*
      DO 140 I = 1, NREG
         WRITE (NOUT,99999) THETA(NPAR+MN+I+1), SE(NPAR+MN+I+1), BX(I)
  140 CONTINUE
*
      NT = 4
      IFLAG = 0
      CALL G13FFF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFLAG)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Volatility forecast = ', CVAR(NT)
      WRITE (NOUT,*)
*
      DIST = 'T'
      MEAN = 3.0D0
*
      DO 160 I = 1, NUM
         FAC1 = DBLE(I)*0.01D0
         X(I,2) = 0.01D0 + 0.7D0*SIN(FAC1)
         X(I,1) = 0.5D0 + FAC1*0.1D0
         X(I,3) = 1.0D0
  160 CONTINUE
*
      MN = 1
      NREG = 2
      GAMMA = 0.09D0
      IP = 1
      IQ = 1
*
      NPAR = IQ + IP + 1
      PARAM(1) = 0.05D0
      PARAM(2) = 0.1D0
      PARAM(3) = 0.8D0
*
      FCALL = .TRUE.
      ISEED(1) = 111
      CALL G05KBF(IGEN,ISEED)
*
      IFLAG = 0
      CALL G05HMF(DIST,200,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
      FCALL = .FALSE.
      IFLAG = 0
      CALL G05HMF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
      IFLAG = 0
      CALL G05HMF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFLAG)
*
      DO 200 I = 1, NUM
         XTERM = ZERO
         DO 180 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
  180    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
  200 CONTINUE
*
      IFLAG = -1
*
      MAXIT = 100
```

```
      TOL = 1.0D-5
*
      DO 220 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5D0
  220 CONTINUE
*
      THETA(NPAR+1) = GAMMA*0.5D0
*
      THETA(NPAR+2) = DF*0.65D0
*
      IF (MN.EQ.1) THEN
         THETA(NPAR+2+MN) = MEAN*0.5D0
      END IF
      DO 240 I = 1, NREG
         THETA(NPAR+2+MN+I) = BX(I)*0.5D0
  240 CONTINUE
*
      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.
      HP = 0.5D0
*
      NPAR2 = 3 + IP + IQ + MN + NREG
*
      CALL G13FEF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,NPAR2,THETA,SE,SC,
     +            COVR,LDCOVR,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WORK,LWORK,
     +            IFLAG)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Students t distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '          Parameter       Standard        Correct'
      WRITE (NOUT,*) '          estimates       errors          values'
*
      DO 260 I = 1, NPAR
         WRITE (NOUT,99999) THETA(I), SE(I), PARAM(I)
  260 CONTINUE
*
      WRITE (NOUT,99999) THETA(NPAR+1), SE(NPAR+1), GAMMA
*
      WRITE (NOUT,99999) THETA(NPAR+2), SE(NPAR+2), DF
*
      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) THETA(NPAR+2+MN), SE(NPAR+2+MN), MEAN
      END IF
*
      DO 280 I = 1, NREG
         WRITE (NOUT,99999) THETA(NPAR+2+MN+I), SE(NPAR+2+MN+I), BX(I)
  280 CONTINUE
*
      NT = 4
      IFLAG = 0
      CALL G13FFF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFLAG)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Volatility forecast = ', CVAR(NT)
      WRITE (NOUT,*)
  300 CONTINUE
*
99999 FORMAT (1X,3F16.3)
99998 FORMAT (1X,A,F12.3)
99997 FORMAT (1X,' ** G05HMF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
G05HMF Example Program Results


Normal distribution

          Parameter       Standard        Correct
          estimates       errors          values
           0.371           0.078           0.400
           0.104           0.026           0.100
           0.708           0.041           0.700
           0.119           0.037           0.100
           4.100           0.095           4.000
           1.425           0.059           1.500
           2.260           0.068           2.500

Volatility forecast =        1.707


Students t distribution

          Parameter       Standard        Correct
          estimates       errors          values
           0.038           0.009           0.050
           0.083           0.025           0.100
           0.811           0.027           0.800
           0.116           0.038           0.090
           5.753           0.625           5.100
           2.968           0.036           3.000
           1.489           0.023           1.500
           2.516           0.030           2.500

Volatility forecast =        0.598
```